Using Strings, String Concatenation and Using String Functions

Stringds is text composed of a sequence of zero or more characters.  A string might contain alphabetic characters, numbers and/or symbols.

Internally (i.e. within the computer), each character of a string is represented by a numeric value called a character code (sometimes referred to as an ASCII code).  Character codes range from 0 to 255.  The use of character codes is important when using or identifying non-printable characters (like the Return key) or characters which are not part of the standard keyboard (like "Ø").

Strings can be of fixed length or variable length.

	'A variable-length string declaration
Dim sVariable As String 

'A fixed-length string declaration
Dim sFixed As String * 10


Fixed-length strings have the characteristic of always containing the number of characters with which they're declared.  In the example above, strFixed will always contain strings of exactly 10 characters, whereas strVariable may contain strings of different lengths.  If a string of fewer than 10 characters is assigned to strFixed, spaces will be added to the end of the string until strFixed contains exactly 10 characters.  However, if a string of more than 10 characters is assigned, only the first 10 will be stored; any remaining characters will be lost.
DEFINTION: A string is a sequence of text of zero or more characters. (The maximum is String length in VB is 2 billion characters) The string can be alphabetic characters, numbers and/or symbols.

Think of a string as a line of characters (numbers, letters and symbols) each is in a certain position. The positions a numbered from 1 to the maximum string length (2 billion-ish).

So if I had a string with “Bert Smith” in it the string would look something like this:

	B
	E
	R
	T
	
	S
	M
	I
	T
	H

	1
	2
	3
	4
	5
	6
	7
	8
	9
	10


Declaring the a Variable of Datatype String

To declare a variable of the datatype string, just use the normal declaration statement:

	Dim strVariableName as String


Putting a String into a String Variable

To put a string into a string variable is easy you can use one or more of the methods below to do it, either putting a string in from a string written in the code, or from another variable or from a control on the form, e.g. a text box.

	strStudentName = “Bert Smith”


	strStudentName = strFirstName


	strStudentName = txtStudentName.Text


Getting a String from a String Variable

Getting information from a string is easy you can take it from the variable and show it in a control, such as a text box or you can put it into another variable.

	txtStudentName.txt = StrStudentName


	strUserInput = StrStudentName


Concatenation

Concatenation, although it sounds painful is a computing term for sticking two or more things (strings) together to form one thing (string).

Concatenating two strings means sticking them together, one after another, giving a new string. For example, the string `foo' concatenated with the string `bar' gives the string `foobar'.

In Visual Basic like any other languages there is a “Concatenation Operator”  that you need to use when sticking together two strings. The Concatenation Operator is the amphasand “&” symbol, to see how you can use it see the code below:

You use it by placing it between the two strings you want to join together these can be both variables or one of them a variable or both string written in your code.

	StrBoth = strFirstString & strSecondString

StrBoth = strFirstString & “Hello”

StrBoth = txtFirstString & “ “ & txtSecond String


Remember you may need to add spaces in your code as shown in the third line of code above.

You can use concatenation almost anywhere in VB for example here is a message box with concatenation of two strings, check the code out below:

	strFirstString = "Hello"

strSecondString = "There"

MsgBox (strFirstString & " " & strSecondString)

Using String Functions

What follows is a brief description of some of the simpler, more common string handling functions that Visual Basic provides. 

Len(string)

Determines the length of string. 

sName = "Basic"
i = Len(sName)  'Returns the value 5. 

Chr$(character_code)

Creates a one-character string from character_code. 

s = Chr$(66)  'Creates "B" 

As demonstrated above, the character code for "B" is 66. 

Asc(string)

Returns the character code of the first character in string. 

i = Asc("Basic")  'Returns the value 66. 

Space$(length)

Creates a string of spaces length characters long. 

sBuffer = Space$(10)  'Returns "          " 

String$(length, string)

Creates a string length characters long that contains only the first character of string.  If string is longer than one character, only the first character of string is used. 

s = String$(10, "ABCD")  'Returns "AAAAAAAAAA" 

LCase$(string)

Converts all alphabetic characters in string to lower case. 

s = LCase$("A1B2c3d4")  'Returns "a1b2c3d4" 

UCase$(string)

Converts all alphabetic characters in string to upper case. 

s = UCase$("A1B2c3d4")  'Returns "A1B2C3D4" 

LTrim$(string)

Removes spaces from the left side of a string. 

s = LTrim$("  ABC  ")  'Returns "ABC  " 

RTrim$(string)

Removes spaces from the right side of string. 

s = RTrim$("  ABC  ")  'Returns "  ABC" 

Trim$(string)

Removes spaces from both sides of string. 

s = Trim$("  ABC  ")  'Returns "ABC" 

StrReverse(string)

NOTE:  This function is new to Visual Basic, version 6.0. 

Reverses the order of all the characters in string. 

s = StrReverse("ABCD")  'Returns "DCBA" 

Using Substrings

Substrings are pieces of a string.  What follows are a few string functions that help you deal with substrings. 

Left$(string, length)

Returns length characters from the left side of string. 

s = Left$("This is a test", 4)  'Returns "This" 

Right$(string, length)

Returns length characters from the right side of string. 

s = Right$("This is a test", 4)  'Returns "test" 

Mid$(string, start[, length])

Mid$ is unique in that it is a command as well as a function.  As a function, it returns a substring from string, beginning at the start character position for length characters.  As a command, it replaces a substring of string, beginning at the start character position for length characters.  If length is not provided, all the characters from start to the end of the string are included in the substring. 

s = Mid$("This is a test", 6, 4)  'Returns "is a"
s = Mid$("This is a test", 6)     'Returns "is a test"

sMid = "This is a test"
Mid$(sMid, 6, 4) = "1234567890"  'Only 4 chars will be used
'sMid now contains "This 1234 test" 

sMid = "This is a test"
Mid$(sMid, 6) = "1234567890"
'sMid now contains "This 123456789" 

sMid = "This is a test"
Mid$(sMid, 6) = "12"
'sMid now contains "This 12 a test" 

These are the most commonly used string functions be sure to revise these as much as possible.

Split(string[, delimiter[, count[, compare_mode]]])

NOTE:  This function is new to Visual Basic, version 6.0. 

Splits string and returns the substrings as an array of strings.  Delimiter identifies where the split(s) should occur.  If delimiter is not provided, a space (" ") is used.  Count is the maximum number of substrings to return.  If count is not provided, all substrings are to be returned.  Compare_mode indicates how the search for delimiter inside string is to be performed.  Refer to the section Comparing Strings for more information about compare_mode. 

Dim i As Integer
Dim sArray As Variant 

sArray = Split("This is a test")
For i = LBound(sArray) To UBound(sArray)
  Debug.Print i; sArray(i)
Next 

'This will display the following:
'  0 This
'  1 is
'  2 a
'  3 test 

Notice that the return array is zero-based (not one-based).  A few more examples follow: 

sArray = Split("This is a test", "s")
'Returns:
'  "Thi"
'  " i"
'  " a te"
'  "t" 

sArray = Split("This is a test", , 2)
'This splits the array like the first example,
'  but only 2 substrings will be returned:
'  "This"
'  "is a test" 

sArray = Split("34,658,-39", ",")
'Returns:
'  "34"
'  "658"
'  "-39" 

Join(string_array[, delimiter])

NOTE:  This function is new to Visual Basic, version 6.0. 

Concatenates the substrings in the array string_array into a single string.  Delimiter is a string that will be concatenated between each of the substrings.  If delimiter is not provided, a space will be used.  This is the opposite of the Split function. 

Dim sArray(4 to 7) As String
Dim sJoin As String 

sArray(4) = "This"
sArray(5) = "is"
sArray(6) = "a"
sArray(7) = "test"
sJoin = Join(sArray, " ")   'Returns "This is a test"
sJoin = Join(sArray)        'Returns "This is a test"
sJoin = Join(sArray, ",")   'Returns "This,is,a,test"
sJoin = Join(sArray, "..")  'Returns "This..is..a..test"
sJoin = Join(sArray, "")    'Returns "Thisisatest" 

Exercises

Here are some exercises to help you learn about strings, concatenation and string functions.

Task 1 – Write a program with a form that has two text boxes called “txtFirstName” and “txtSurname”, also include a command button called “cmdClickMe”. When the command button is pressed the program should show two message boxes one that shows the “strFirstName” variable and the other shows the “strSurname” variable.

Task 2 – Adjust the above program to include another text box called “txtFullName” remove the code from the command button and then write new code the will take the txtFirstname and the txtSurname and “Concatenate” (&) them together, then display the full name in txtFullName.

Task 3 – Start a new program create a form with the following controls:

1 Text Box

· txtUserInput

5 Label Boxes

· lblLengthOfString

· lblLowerCaseString

· lblUpperCaseString

· lblLeftOfString

· lblRightOfString

· lblMidOfString

1 Command Button

· cmdClickMe

The program should take the users input as a string and place it in a variable called “strUserInput”, it should then perform the following functions on the string variable (strUserInput) and show the result in each of the corresponding label boxes. The functions that should be performed are:

· Length

· Lcase$

· Ucase$

· Left$

· Right$

· Mid$

You can choose your own start, end and lengths for the Left, Right and Mid functions.

